
1 Introduction

As part of implementing a compiler from Aquery to q1, we have developed
the BNF grammar below. This grammar is eventually used in implementing a
flex/bison parser for Aquery.

2 Grammar

An Aquery program consists on a top level composed by queries, table/view
creation, insert/update/delete statements. An empty program is also a valid
aquery program as noted by the epsilon production for top-level.

2.1 Top-level program

〈program〉 ::= 〈top level〉

〈top level〉 ::= 〈global query〉 〈top level〉
| 〈create table or view〉 〈top level〉
| 〈insert statement〉 〈top level〉
| 〈update statement〉 〈top level〉
| 〈delete statement〉 〈top level〉
| 〈user function definition〉 〈top level〉
| ε (*note: an empty aquery program is still an aquery program*)

2.2 Local and global queries

We proceed to define what constitutes a global and local query, those that can
solely be used within queries between the WITH keyword and the following
global query. Note that aside from the necessary declarations at the beginning,
the remainder of the query is a normal query, and thus refers to the grammar
rule associated with the base query non-terminal.

〈global query〉 ::= 〈local queries〉 〈base query〉
1Aquery is an ordered database query language developed by Alberto Lerner and Dennis

Shasha, for more information please see https://cs.nyu.edu/web/Research/TechReports/

TR2003-836/TR2003-836.pdf

〈local queries〉 ::= ’WITH’ 〈local query〉 〈local queries tail〉
| ε

〈local queries tail〉 ::= 〈local query〉 〈local queries tail〉
| ε

〈local query〉 ::= 〈identifier〉 〈col aliases〉 ’AS’ ’(’ 〈base query〉 ’)’

〈col aliases〉 :: = ’(’ 〈comma identifier list〉 ’)’
| ε

〈comma identifier list〉 ::= 〈identifier〉 〈comma identifier list tail〉

〈comma identifier list tail〉 ::= ’,’ 〈identifier〉 〈comma identifier list tail〉
| ε

〈column list〉 ::= 〈column elem〉 〈column list tail〉

〈column elem〉 ::= 〈identifier〉 | 〈column access〉

〈column access〉 ::= 〈identifier〉 ’.’ 〈identifier〉

〈column list tail〉 ::= ’,’ 〈column elem〉 〈column list tail〉
| ε

2.3 Base query

A query requires a select clause and a from clause. There are additional optional
clauses including an ordering clause (the base of declarative order in Aquery), a
where clause, a group by clause, and a having clause, which can be used solely
in conjunction with a group by clause.

〈base query〉 ::= 〈select clause〉 〈from clause〉 〈order clause〉 〈where clause〉
〈groupby clause〉

〈select clause〉 ::= ’SELECT’ 〈select elem〉 〈select clause tail〉

〈select elem〉 ::= 〈value expression〉 ’as’ 〈identifier〉
| 〈value expression〉

〈select clause tail〉 ::= ’,’ 〈select elem〉 〈select clause tail〉
| ε

〈from clause〉 ::= ’FROM’ 〈table expressions〉

〈order clause〉 ::= ’ASSUMING’ 〈order specs〉
| ε

〈order specs〉 ::= 〈order spec〉 〈order specs tail〉

〈order spec〉 ::= ’ASC’ 〈column list〉
| ’DESC’ 〈column list〉

〈order specs tail〉 ::= ’AND’ 〈order spec〉 〈order specs tail〉
| ε

〈where clause〉 ::= ’WHERE’ 〈search condition〉
| ε

〈groupby clause〉 ::= ’GROUP’ ’BY’ 〈comma value expression list〉 〈having clause〉
| ε

〈having clause〉 ::= ’HAVING’ 〈search condition〉
| ε

2.3.1 Search condition

where clauses, along with other clauses such as having and on, require a search
condition non-terminal. We borrow the notion of a search condition from SQL
92 grammar, but also allow traditional value expressions to be a member of a
search condition, which allows user defined functions to be used as stand-alone
predicates, as well as boolean-typed columns. Please see http://savage.net.

au/SQL/sql-92.bnf.html#searchcondition for the SQL92 details on this.

〈search condition〉 ::= 〈boolean term〉
| 〈search condition〉 ’OR’ 〈boolean term〉

〈boolean term〉 ::= 〈boolean factor〉
| 〈boolean term〉 ’AND’ 〈boolean factor〉

〈boolean factor〉 ::= 〈boolean primary〉
| ’NOT’ 〈boolean primary〉

〈boolean primary〉 ::= 〈predicate〉
| ’(’ 〈search condition〉 ’)’

〈predicate〉 ::= 〈value expression〉 〈posfix predicate〉
| 〈overlaps predicate〉

〈posfix predicate〉 ::= 〈between predicate〉
| 〈in predicate〉
| 〈like predicate〉
| 〈null predicate〉
| 〈is predicate〉
| ε (* value expression is already boolean *)

〈between predicate〉 ::= BETWEEN 〈value expression〉 ’AND’ 〈value expression〉
| ’NOT’ ’BETWEEN’ 〈value expression〉 ’AND’ 〈value expression〉

〈in predicate〉 ::= ’IN’ 〈in pred spec〉
| ’NOT’ ’IN’ 〈in pred spec〉

〈in pred spec〉 ::= 〈value expression〉 (* implicit list *)
| ’(’ 〈comma value expression list〉 ’)’

〈like predicate〉 ::= ’LIKE’ 〈value expression〉
| ’NOT’ ’LIKE’ 〈value expression〉

〈null predicate〉 ::= ’IS’ ’NULL’
| ’IS’ ’NOT’ ’NULL’

〈is predicate〉 ::= ’IS’ 〈truth value〉
| ’IS’ ’NOT’ 〈truth value〉

〈truth value〉 ::= ’TRUE’ | ’FALSE’

〈overlaps predicate〉 ::= 〈range value expression〉 ’OVERLAPS’ 〈range value expression〉

〈range value expression〉 ::= ’(’ 〈value expression〉 ’,’ 〈value expression〉 ’)’

2.3.2 Table Expressions

We now proceed to define what table expressions constitute. Informally, ta-
ble expression can be an identifier associated with a table, or an operation
on a table (such as flatten), or a join on tables. Join grammar inspired by
http://savage.net.au/SQL/sql-92.bnf.html#joinedtable. Note that the
precedence/associativity of joins and other operations is directly encoded in the
grammar.

〈table expressions〉 ::= 〈joined table〉 〈table expressions tail〉

〈table expressions tail〉 ::= ’,’ 〈joined table〉 〈table expressions tail〉 (*note
the semantics of this are cross join*)

| ε

〈joined table〉 ::= 〈qualified join〉
| 〈qualified join〉 ’CROSS’ ’JOIN’ 〈joined table〉

〈qualified join〉 ::= 〈table expression〉
| 〈table expression〉 ’NATURAL’ 〈join type〉 ’JOIN’ 〈qualified join〉
| 〈table expression〉 〈join type〉 ’JOIN’ 〈qualified join〉 〈join spec〉

〈table expression〉 ::= 〈table expression main〉
| 〈built in table fun〉 ’(’ 〈table expression main〉 ’)’

〈table expression main〉 ::= 〈identifier〉 〈identifier〉 (* implicit correlation name
*)

| 〈identifier〉 ’AS’ 〈identifier〉 (* explicit correlation name *)
| 〈identifier〉
| ’(’ 〈joined table〉 ’)’

〈built in table fun〉 ::= ’FLATTEN’ (* potentially more to add here *)

〈join type〉 ::= ’INNER’ | ’LEFT’ ’OUTER’ | ’LEFT’ | ’RIGHT’ ’OUTER’ |
’RIGHT’ | ’FULL’ | ’FULL’ ’OUTER’ | ε

〈join spec〉 ::= 〈on clause〉 | 〈using clause〉

〈on clause〉 ::= ’ON’ 〈search condition〉

〈using clause〉 ::= ’USING’ ’(’ 〈comma identifier list〉 ’)’

2.4 Table and View Creation

We define table and view creation at the top-level

〈create table or view〉 ::= ’CREATE’ ’TABLE’ ’ID’ 〈create spec〉
| ’CREATE’ ’VIEW’ ’ID’ 〈create spec〉

〈create spec〉 ::= ’AS’ 〈global query〉
| ’(’ 〈schema〉 ’)’

〈schema〉 ::= 〈schema element〉 〈schema tail〉

〈schema element〉 ::= 〈identifier〉 〈type〉

〈schema tail〉 ::= ’,’ 〈schema element〉 〈schema tail〉
| ε

〈type〉 ::= ’INT’ |’FLOAT’ |’STRING’ | ’DATE’ | ’BOOLEAN’ | ’HEX’

2.5 Updating, Inserting, Deleting

We define the grammar relating to update, insert and delete statements at the
top level.

〈update statement〉 ::= ’UPDATE’ 〈identifier〉 ’SET’ 〈set clauses〉 〈order clause〉
〈where clause〉

〈set clauses〉 ::= 〈set clause〉 〈set clauses tail〉

〈set clauses tail〉 ::= ’,’ 〈set clause〉 〈set clauses tail〉
| ε

〈set clause〉 ::= 〈identifier〉 ’=’ 〈value expression〉

〈insert statement〉 ::= ’INSERT’ ’INTO’ 〈identifier〉 〈order clause〉 〈insert modifier〉
〈insert source〉

〈insert modifier〉 ::= ’(’〈comma identifier list〉 ’)’ (* insert values into given
order of column names *)

| ε (* insert into default column order *)

〈insert source〉 ::= 〈global query〉
| ’VALUES’ ’(’ 〈comma value expression list〉 ’)’

〈delete statement〉 ::= ’DELETE’ 〈from clause〉 〈order clause〉 〈where clause〉
| ’DELETE’ 〈comma identifier list〉 〈from clause〉 (* similarly to q, we can

choose to delete rows where the predicates in where clause are true, or we
can choose to delete a column, but not both. No need to specify order,
since will delete whole column...*)

2.6 User Defined Functions

We now define another element of the top-level: user defined function. Functions
can have a series of expressions, queries, or local variable definitions. All but
the last of which have to be followed by a semi-colon. The result of the function
is the last expression evaluated.

〈user function definition〉 ::= ’FUNCTION’ 〈identifier〉 ’(’ 〈comma identifier list〉
’)’ ’{’ 〈function body〉 ’}’

〈function body〉 ::= 〈function body elem〉 〈function body tail〉
| ε (*note: a function with no body is still a function *)

〈function body tail〉 ::= ’;’ 〈function body elem〉 〈function body tail〉
| ε

〈function body elem〉 ::= 〈value expression〉 | 〈function local var def 〉 | 〈local queries〉
〈base query〉 (* functions can have expressions or queries *)

〈function local var def 〉 ::= 〈identifier〉 ’:=’ 〈value expression〉

2.7 Value Expressions

We encode operator precedence and associativity into the grammar itself. This
section of the grammar draws inspiration from http://www.lysator.liu.se/

c/ANSI-C-grammar-y.html. Some expressions also draw inspiration from http:

//savage.net.au/SQL/sql-92.bnf.html.

〈constant〉 ::= 〈integer〉 | 〈float〉 | 〈date〉 | 〈string〉 | 〈hex 〉 | 〈truth value〉

〈table constant〉 ::= ’ROWID’ | 〈column access〉 | ’*’

〈case expression〉 ::= ’CASE’ 〈case clause〉 〈when clauses〉 〈else clause〉 ’END’
(* sql92 *)

〈case clause〉 ::= 〈value expression〉 (* value is compared to values in when
clauses*)

| ε (* when clause is treated as a search *)

〈when clauses〉 ::= 〈when clause〉 〈when clauses tail〉

〈when clauses tail〉 ::= 〈when clause〉 〈when clauses tail〉
| ε

〈when clause〉 ::= ’WHEN’ 〈search condition〉 ’THEN’ 〈value expression〉 (*search
condition generalizes to the value-comparison case expression, since a search
condition non-terminal can expand into a simple value expression *)

〈else clause〉 ::= ’ELSE’ 〈value expression〉
| ε

〈main expression〉 ::= 〈constant〉 | 〈table constant〉 | 〈identifier〉 | ’(’ 〈value expression〉
’)’ | 〈case expression〉

〈call〉 ::= 〈main expression〉
| 〈main expression〉 ’[’ 〈indexing〉 ’]’
| 〈built in fun〉 ’(’ ’)’
| 〈built in fun〉 ’(’ 〈comma value expression list〉 ’)’
| 〈identifier〉 ’(’ ’)’ (* user defined functions *)
| 〈identifier〉 ’(’ 〈comma value expression list〉 ’)’ (*user defined functions*)

〈indexing〉 ::= ODD | EVEN | EVERY 〈integer〉

〈built in fun〉 ::= ’abs’ | ’avg’ | ’count’ | ’deltas’ | ’distinct’ | ’drop’ | ’fill’ |
’first’ | ’last’ | ’max’ | ’maxs’ | ’min’ | ’mins’ | ’mod’ | ’next’ | ’not’| ’prev’ |
’prd’ | ’prds’ | ’reverse’ | ’sum’ | ’sums’ | ’stddev’|

〈exp expression〉 ::= 〈call〉
| 〈call〉 〈exp op〉 〈exp expression〉

〈mult expression〉 ::= 〈exp expression〉
| 〈mult expression〉 〈times op〉 〈exp expression〉
| 〈mult expression〉 〈div op〉 〈exp expression〉

〈add expression〉 ::= 〈mult expression〉
| 〈add expression〉 〈plus op〉 〈mult expression〉
| 〈add expression〉 〈minus op〉 〈mult expression〉

〈rel expression〉 ::= 〈add expression〉
| 〈rel expression〉 〈l op〉 〈add expression〉
| 〈rel expression〉 〈g op〉 〈add expression〉
| 〈rel expression〉 〈le op〉 〈add expression〉
| 〈rel expression〉 〈ge op〉 〈add expression〉

〈eq expression〉 ::= 〈rel expression〉
| 〈eq expression〉 〈eq op〉 〈rel expression〉
| 〈eq expression〉 〈neq op〉 〈rel expression〉

〈and expression〉 ::= 〈eq expression〉
| 〈and expression〉 〈and op〉 〈eq expression〉

〈or expression〉 ::= 〈and expression〉
| 〈or expression〉 〈or op〉 〈and expression〉

〈value expression〉 ::= 〈or expression〉

Now that we have value expressions defined, we define a form of value ex-
pression list: comma separated value expressions.

〈comma value expression list〉 ::= 〈value expression〉 〈comma value expression list tail〉

〈comma value expression list tail〉 ::= ’,’ 〈value expression〉 〈comma value expression list tail〉
| ε

This concludes the formal outline of the Aquery grammar. Note that this
grammar maybe revised and changed as necessary throughout development if
need be.

For a flex/bison implementation of this grammar please see https://www.

github.com/josepablocam/aquery2q/parser/

